6642126

DP-1098

see Rev. 1

AEC RESEARCH AND DEVELOPMENT REPORT

RADOS, A Code to Estimate Gamma Dose from a Cloud of Radioactive Gases

R. E. Cooper

DUPONT
REG. U. S. PAT. OFF.

Savannah River Laboratory

Aiken, South Carolina

LEGAL NOTICE

Printed in the United States of America
Available from
Clearinghouse for Federal Scientific and Technical Information
National Bureau of Standards, U. S. Department of Commerce
Springfield, Virginia 22151
Price: Printed Copy $3.00; Microfiche $0.65

664212

DP-1098

Health and Safety
(TID-4500)

RADOS, A Code to Estimate Gamma Dose from a Cloud of Radioactive Gases

by

Roger E. Cooper

Approved by

E. C. Nelson, Research Manager
Reactor Engineering Division

June 1967

E. I. DU PONT DE NEMOURS & COMPANY
SAVANNAH RIVER LABORATORY
AIKEN, S. C. 29801

CONTRACT AT(07-2)-1 WITH THE
UNITED STATES ATOMIC ENERGY COMMISSION

ABSTRACT

RADOS is a computer code that represents the finite spatial distribution of airborne source material as an infinite number of line sources. The code provides a means of rapidly calculating whole body gamma dose from a finite cloud of radioactive material. The simplifying assumptions used to minimize computation time are:

- The material release is instantaneous and results in a radioactive cloud of unit thickness in the X or downwind direction.
- There is no change in the size and shape of the cloud during passage over the effective range of the receptor.
- The gamma buildup factors can be expressed analytically with sufficient accuracy.

These assumptions limit the applicability of RADOS as follows: the material release should occur over a relatively short time; the receptor distance should be more than 600 meters downwind from the release point; and the gamma energies should be in the specified range.

A 4-group, 12-isotope, 1-receptor-point problem requires approximately 1.75 seconds on the IBM System/360-65.

INTRODUCTION

Hazards analyses associated with the design and operation of nuclear plants frequently require the study of postulated accidents involving the release of radioactive material to the atmosphere. Individuals downwind from this postulated release could be subjected to (1) an inhalation dose if enveloped by the cloud of material and (2) an external or whole body gamma dose even if there were no contact with the cloud. Because dispersion conditions are likely to vary widely at a particular site, a parametric survey may be useful in estimating doses. This usually requires a large number of machine computations. Inhalation doses may be readily and economically estimated by calculations with existing codes. However, available methods for calculating external gamma doses either over-simplified the problem (such as the semi-infinite cloud method) or required excessive running time for the more valid finite cloud method. The RADOS code was therefore written to provide a fast running and reasonably accurate gamma dose calculation utilizing a finite cloud model. The code is based on a mathematical model developed by L. M. Arnett.[1]

SUMMARY

RADOS has been used to evaluate external gamma doses for postulated releases at the Savannah River Plant. The simplifying assumptions were limiting only for very special and infrequently encountered cases. RADOS calculations agree well with the more detailed CLOUD[2] calculations as shown in Table I. The running time for RADOS is approximately 8% of that required for CLOUD.

A FORTRAN IV listing of the code, including the required Bessel function routines, is included in Appendix A.

TABLE I

Comparative Calculations

Conditions	Relative Dose CLOUD	RADOS
Stable	100.0	98.0
↓	99.7	97.0
	95.7	96.3
	87.0	75.7
	50.3	44.0
Unstable	22.8	23.2

ANALYTIC DESCRIPTION

The principal assumption in this calculational method is that there is no change in the size and shape of the cloud of unit thickness during the time the cloud will irradiate the receptor. Beyond 600 meters from the point of release, this assumption introduces negligible error for meteorological conditions of interest. For a cloud of constant size and shape during the exposure interval, the entire dose may be considered as being accumulated in unit time from a cloud extending to infinity in both directions from the receptor. This assumption permits the total dose to be calculated as the summation of the doses from an infinite number of line sources.

The dose from a line source at a distance "a" from the receptor is

$$D(a) = \frac{\nu}{4\bar{u}} \frac{S(a)G(\mu a)}{a} \qquad (1)$$

where

ν = factor to convert source intensity in units* such as Mev/s to intensity in units such as roentgens per unit time

$\bar{u}$ = average wind velocity

$S(a)$ = source per unit distance at "a"

$G(\mu a)$ = attenuation of the radiation due to air from the line source at the distance "a"

μ = linear attenuation coefficient of air for gamma radiation

The total dose is then

$$D = \frac{S\nu}{4\bar{u}} \int_0^\infty \frac{f(a)G(\mu a)da}{a} \qquad (2)$$

* Although units in meter-seconds are specified in the code description, any set that is consistent may be used.

where

$$S = \int_0^\infty S(a)da, \text{ the total source}$$

$f(a)$ = relative source strength in the cloud, and

$$\int_0^\infty f(a)da = 1$$

This total dose D is the composite of doses from the individual isotopes, i, in the source or

$$D = \sum_i \frac{S_i \nu_i \exp(-\lambda_i t)}{4\bar{u}} \int_0^\infty \frac{f(a)G(\mu_i a)da}{a} \qquad (3)$$

S_i is the source strength at time of release and "t" is some average time to allow for decay of the source during the passage from release point to receptor.

The function $G(\mu a)$, the attenuation due to air of the radiation emitted from the line source, is

$$G(\mu a) = \frac{2}{\pi} \int_0^{\pi/2} B \exp(-\mu a \sec\phi)d\phi \qquad (4)$$

The buildup factor B in air for gamma energies from 0.5 to 2 Mev has been taken as

$$B = 1 + \mu a + \frac{(\mu a)^2}{7E^{2.4}} \qquad (5)$$

where E is the energy of the gamma photon in Mev. For this value of B,

$$G(\mu a) = \frac{2}{\pi} \left\{ K_{i1}(\mu a) + \mu a K_o(\mu a) + \frac{(\mu a)^2}{7E^{2.4}} K_1(\mu a) \right\} \qquad (6)$$

where the K's are Bessel functions.

The space distribution function f(a) can be chosen to fit the meteorological conditions of interest. Sutton[3] and many others[4] have assumed that the concentration in the cloud is Gaussian both vertically and horizontally. When this form of f(a) is assumed,

$$f(a)da = \frac{1}{\pi}\alpha^2\beta\gamma d\gamma\int_0^{\pi} \exp\left\{-\tfrac{1}{2}\alpha^2\left[\beta^2\gamma^2\sin^2\theta + (\gamma\cos\theta - 1)^2\right]\right\}d\theta \qquad (7)$$

where

h = height of the center of concentration above the ground

σ_z, σ_y = Gaussian constants of the concentration in the vertical and horizontal directions, respectively

$\alpha = h/\sigma_z$

$\beta = \sigma_z/\sigma_y$

$\gamma = a/h$

Also

$$\frac{f(a)da}{a} = \frac{F(\gamma)d\gamma}{h} \qquad (8)$$

where

$$F(\gamma) = \frac{1}{\pi}\alpha^2\beta\int_0^{\pi} \exp\left\{-\tfrac{1}{2}\alpha^2\left[\beta^2\gamma^2\sin^2\theta + (\gamma\cos\theta - 1)^2\right]\right\}d\theta$$

Equation (3) now becomes

$$D = \sum_i \frac{S_i\nu_i \exp(-\lambda_i t)}{4\bar{u}h}\int_0^{\infty} F(\gamma)G(\mu_i h\gamma)d\gamma \qquad (9)$$

The expression before the integral in equation (9) is the dose that the receptor would receive if all the activity in the cloud were concentrated in a line source at the distance h. The integral is a correction factor for the absorption of the gamma radiation in the air and for the distribution of the activity in space. The value of the integral is near unity for most cases of interest. It is always true that

$$\int_0^{\infty} F(\gamma)d\gamma = 1$$

Another space distribution of interest is "fumigation" which arises when the atmosphere is near neutral at the ground but an inversion is based at some higher level. To simulate these atmospheric conditions, the concentration of the activity is assumed to be homogeneous vertically and Gaussian horizontally.

The expression that corresponds to equation (9) is

$$D = \sum_i \frac{S_i \nu_i \exp(-\lambda_i t)}{4\bar{u}\ell} \left[\int_0^1 F_1(\gamma) G(\mu_i \ell \gamma) d\gamma + \int_1^\infty F_2(\gamma) G(\mu_i \ell \gamma) d\gamma \right] \quad (10)$$

where

ℓ = height of inversion layer or lid

$$F_1(\gamma) = \sqrt{\frac{2}{\pi}}\, \beta \int_0^{\pi/2} \exp(-\tfrac{1}{2}\beta^2\gamma^2 \sin^2\theta) d\theta$$

$$F_2(\gamma) = \sqrt{\frac{2}{\pi}}\, \beta \int_{\text{arcsec}\gamma}^{\pi/2} \exp(-\tfrac{1}{2}\beta^2\gamma^2 \sin^2\theta) d\theta$$

$\beta = \ell/\sigma_y$

$\gamma = a/\ell$

CALCULATIONAL PROCEDURE

RADOS presently accepts meteorological data in either of two forms.

- Sutton's diffusion parameters C_y, C_z, and n, the horizontal and vertical diffusion coefficients and stability parameter, respectively.
- Generalized Gaussian dispersion parameters σ_y and σ_z.

The relationships between these two sets of parameters are

$$\sigma_y = \frac{C_y X^{(2-n)/2}}{\sqrt{2}} \quad \text{and} \quad \sigma_z = \frac{C_z X^{(2-n)/2}}{\sqrt{2}} \quad (11)$$

where X is the distance from the release point to the receptor. The code determines the form of input by testing for the presence of n, the stability parameter in Sutton's equation. This parameter will be entered as zero or left blank if the generalized Gaussian dispersion parameters are used. In either case, the appropriate calculational procedure will be automatically selected. The results shown in Table II were obtained by solving equation (9) with Sutton's diffusion parameters as input.

TABLE II

Dose Calculations by CLOUD and RADOS

Release height = 70 meters

Wind speed = 3 meters/second

Distance X = 8×10^3 meters

Sutton's Parameters			Relative Dose	
C_y, meters$^{n/2}$	C_z, meters$^{n/2}$	n	CLOUD	RADOS
0.060	0.030	0.43	100.0	98.0
0.060	0.030	0.38	99.7	97.0
0.064	0.036	0.34	95.7	96.3
0.076	0.056	0.30	87.0	75.7
0.094	0.086	0.26	50.3	44.0
0.116	0.130	0.23	22.8	23.2

The equation for normal Gaussian distribution, in both the horizontal and vertical plane, is solved if the inversion lid parameter is omitted or set to zero. If a lid is specified, the distribution of material is treated as being homogeneous in the vertical plane and Gaussian in the horizontal plane. Both equations are integrated numerically by a double Gaussian quadratures technique.

The dimensionless parameters α and β are generated internally for substitution into the differential equations. A parametric survey of the dose integrals may be made by simply using the σ_y, σ_z option and adjusting ratios.

Total concentration of radioactive isotopes at each detector position is determined by a simple parent-daughter decay scheme:

$$S_{p,i}(t) = S_{o,p,i} \exp(-\lambda_{p,i} t) \qquad (12)$$

$$S_{d,i}(t) = S_{o,d,i} \exp(-\lambda_{d,i} t)$$

$$+ \frac{S_{o,p,i} \lambda_{f,i} E_{d,i,k}}{(\lambda_{d,i} - \lambda_{p,i}) E_{p,i,k}} [\exp(-\lambda_{p,i} t) - \exp(-\lambda_{d,i} t)] \qquad (13)$$

where

$S_{p,i}$ = parent activity of i^{th} isotope, Mev/s

$S_{d,i}$ = daughter activity of i^{th} isotope, Mev/s

$S_{o,p,i}$ = total release inventory of parent activity for i^{th} isotope, Mev/s

$S_{o,d,i}$ = total release inventory of daughter activity for i^{th} isotope, Mev/s

λ_i = decay constant for i^{th} isotope, $seconds^{-1}$

$E_{i,k}$ = gamma energy of the k^{th} energy group of the i^{th} isotope, Mev

t = cloud transit time from release point to downwind receptor = $x/\bar{u}$

Input Preparation

All input information for RADOS is entered by the 6E12.6 format with the exception of the header card for each set of problems. The header card is a 72 column alphameric field that must be supplied with identification information to be used as a title for the output data for each set of problems. A set is a group of problems that use the same isotopic inventory, gamma energy group structure, and downwind receptor points.

A parametric survey of calculated dose as a function of wind speed, release height, and dispersion conditions is performed with minimum preparation since each problem requires only one card of the set. Any number of problems may be included in a set.

If the dispersion parameters are entered in the form of Sutton's C_y's and C_z's, the program will calculate a dose at each downwind receptor point for the dispersion conditions on each problem. If the parameters are entered as σ_y's and σ_z's, a new problem will be read in for each receptor point. Therefore, it is necessary that the parameters for a given set of problems be consistent in form.

Following the identification card, all input should be entered in the order as listed below:

Card Type		Mnemonic	Definition
2	a)	GP	No. of energy groups $1 \leq GP \leq 4$
	b)	SI	No. of isotopes $1 \leq IS \leq 20$
	c)	DIST	No. of downwind receptors $1 \leq DIST \leq 12$
	d)	PROB	No. of problems in this set
3	a)	GNU_1	Dose conversion factor, Group 1, $\frac{r}{hr} / \frac{Mev}{m^2\text{-seconds}}$
	b)	GMU_1	Linear attenuation coefficient, Group 1, m^{-1}
	c)	EN_1	Gamma energy, Group 1, Mev $0.5 \leq Mev \leq 2.0$
	d)	GNU_2	Group 2 constants
	e)	GMU_2	
	f)	EN_2	

There may be two Type 3 cards with up to four sets of constants (Groups = GP, Card Type 2a).

4	a)	P_1	Parent source activity, Mev/s
	b)	PL_1	Parent decay constant, $seconds^{-1}$
	c)	DEP_1	Parent energy group number
	d)	D_1	Daughter source activity, Mev/s
	e)	DL_1	Daughter decay constant, $seconds^{-1}$
	f)	DED_1	Daughter energy group number

One Type 4 card must be supplied for each isotope.

5	a)	XD_1	Downwind receptor distance, m
	b)	XD_2	Up to 2 cards and 12 distances (XD = DIST, Card 2c)
	c)	XD_3	
	d)	XD_4	
	e)	XD_5	
	f)	XD_6	

Card Type	Mnemonic	Definition
6 a)	STKHT	Release height, m
b)	CY	C_y or σ_y, $m^{n/2}$ or m
c)	CZ	C_z or σ_z, $m^{n/2}$ or m
d)	STP	Sutton's stability parameter n, dimensionless
e)	UBAR	Average wind speed, m/s
f)	FLID	Inversion lid height; zero or leave blank for no inversion, m

The number of Type 6 cards should be equal to PROB on Card 2d.

Other sets of problems may be run sequentially by repeating the header card and the above requirements.

Output

Output from each set of problems is identified by the information supplied by the header card for the set. Gamma energy group constants and isotope parameters, as supplied by input data, are printed out after the identification on the first page only. The output for each problem of the set will be one data line of 12 parameters with an appropriate column heading for each parameter. Computed integrals by gamma energy groups are edited to provide information independent of the isotopic content of the cloud; i.e., they are geometric considerations pertaining only to the size and shape of the radioactive cloud.

The input and output for a sample problem are given in Appendix B.

REFERENCES

1. L. M. Arnett. "Calculation of Radiation Dose from a Cloud of Radioactive Gases." Nuclear Applications 3, 217-221 (1967).

2. D. S. Duncan. CLOUD - An IBM 709 Program for Computing Gamma-Ray Dose Rate from a Radioactive Cloud. Atomics International, NAA-SR-Memo-4822 (1959).

3. O. G. Sutton. Micrometeorology. McGraw-Hill Book Co., New York, 1953.

4. F. A. Gifford, Jr. "Atmospheric Dispersion Calculations Using the Generalized Gaussian Plume Method." Nuclear Safety 2(2), 56-59, December 1960.

APPENDIX A

```
C      RADOS MAIN PROGRAM
C
       DIMENSION GNU(4),GMU(4)
       DIMENSION EN(4),P(20)
       DIMENSION PL(20),DEP(20)
       DIMENSION DED(20),WORDS(18)
       DIMENSION D(20),DL(20)
       DIMENSION X(8),H(8)
       DIMENSION SUMGP(4),DOSP(20)
       DIMENSION DOSD(20),SP(20)
       DIMENSION SD(20),TIME(12)
       DIMENSION XD(12)
       COMMON STKHT,PI,UPLIM,ALPH,BETA,GMU,X,H,EN,SUMGP,NG,FLID
C
C
     1 FORMAT (6E12.8)
     2 FORMAT (18A4)
     3 FORMAT (1H1,18A4)
     4 FORMAT (6X,8HSTACK HT,2X,7HCY-SIGY,3X,7HCZ-SIGZ,6X,1HN,7X,4HUBAR,5
      1X,8HGP-1 INT,2X,8HGP-2 INT,2X,8HGP-3 INT,2X,8HGP-4 INT,4X,4HDOSE,3
      1X,8HDISTANCE,//)
     5 FORMAT (1X,I3,11(1X,1PE9.3))
    26 FORMAT (1X,17HRELEASE INVENTORY,2X,12HPARENT,MEV/S,3X,14HDECAY CON
      1ST,/S,3X,12HENERGY GROUP,3X,14HDAUGHTER,MEV/S,2X,14HDECAY CONST,/S
      1,3X,12HENERGY GROUP,/)
     7 FORMAT (//)
    27 FORMAT (12X,I3,3X,6(1PE14.6,2X))
    30 FORMAT (1X,22HENERGY GROUP CONSTANTS,3X,9HDOSE CONV,6X,11HATTEN CO
      1EFF,6X,12HENERGY,MEV/S,/)
    31 FORMAT (18X,I3,3X,3(1PE14.6))
C
       NIN=5
       NOUT=6
       CALL EFTM(64)
       PI=3.141593
       X(1)=.9894009E+0
       X(2)=.9445750E+0
       X(3)=.8656312E+0
       X(4)=.7554044E+0
       X(5)=.6178762E+0
       X(6)=.4580168E+0
       X(7)=.2816036E+0
       X(8)=.9501251E-1
       H(1)=.2715246E-1
       H(2)=.6225352E-1
       H(3)=.9515851E-1
       H(4)=.1246290E+0
       H(5)=.1495960E+0
       H(6)=.1691565E+0
       H(7)=.1826034E+0
       H(8)=.1894506E+0
       L=0
       SQ2=SQRT(2.)
```

```
23 READ (NIN,2) (WORDS(I),I=1,18)
   READ (NIN,1)GP,SI,DIST,PROB
   NG=GP+0.1
   IS=SI+0.1
   NDIST=DIST+0.1
   NPRO=PROB+.1
   KEY=0
   NP=0
   DO 24 N=1,4
24 SUMGP(N)=0.
   READ (NIN,1)(GNU(I),GMU(I),EN(I),I=1,NG)
   READ (NIN,1)(P(I),PL(I),DEP(I),D(I),DL(I),DED(I),I=1,IS)
   READ (NIN,1)(XD(I),I=1,NDIST)
   WRITE (NOUT,7)
42 WRITE (NOUT,3) (WORDS(I),I=1,18)
   WRITE (NOUT,7)
   LINES=5
   IF (KEY)44,44,22
44 WRITE (NOUT,26)
   DO 28 I=1,IS
28 WRITE (NOUT,27)I,P(I),PL(I),DEP(I),D(I),DL(I),DED(I)
   WRITE (NOUT,7)
   WRITE (NOUT,30)
   DO 29 I=1,NG
29 WRITE (NOUT,31)I,GNU(I),GMU(I),EN(I)
   WRITE (NOUT,7)
   WRITE (NOUT,4)
   LINES=IS+NG+12
 9 N=0
10 READ (NIN,1)STKHT,CY,CZ,STP,UBAR,FLID
   IF(STP.NE.0) NPROB=NPRO*NDIST
   NPROB=NPRO
12 N=N+1
   NP=NP+1
   IF (STP)11,11,32
11 SIGZ=CZ
   SIGY=CY
   GO TO 13
32 XD2=XD(N)**((2.-STP)/2.)
   SIGY=XD2*CY/SQ2
   SIGZ=XD2*CZ/SQ2
13 ALPH=STKHT/SIGZ
   BETA=SIGZ/SIGY
   UPLIM=1.25+3.75/ALPH
   TIME(N)=XD(N)/UBAR
   DO 14 NI=1,IS
   NDP=DEP(NI)+0.1
   NDD=DED(NI)+0.1
   PLT=EXP(-PL(NI)*TIME(N))
   DLT=EXP(-DL(NI)*TIME(N))
   SP(NI)=P(NI)*PLT
14 SD(NI)=D(NI)*DLT+(P(NI)*(EN(NDD)*DL(NI)/(EN(NDP)*(DL(NI)-PL(NI))))
  1*(PLT-DLT))
   CONST=TIME(N)/(4.*XD(N))
   IF (FLID)15,15,16
15 CALL RADOS1
   CONST=CONST/STKHT
   GO TO 17
```

```
16     BETA=FLID/SIGY
       CALL RADOS2
       CONST=CONST/FLID
   17  DOSE=0.
       DO 18 NI=1,IS
       NDP=DEP(NI)+0.1
       NDD=DED(NI)+0.1
       DOSP(NI)=SP(NI)*GNU(NDP)*SUMGP(NDP)
       DOSD(NI)=SD(NI)*GNU(NDD)*SUMGP(NDD)
   18  DOSE=DOSE+DOSP(NI)+DOSD(NI)
       DOSE=DOSE*CONST
       L=NP
       NT=4
       IF (STP)19,19,20
   19  WRITE (NOUT,5)L,STKHT,SIGY,SIGZ,STP,UBAR,(SUMGP(I),I=1,NT),DOSE,XD
      1(N)
       GO TO 21
   20  WRITE (NOUT,5)L,STKHT,CY,CZ,STP,UBAR,(SUMGP(I),I=1,NT),DOSE,XD(N)
   21  CONTINUE
       LINES=LINES+1
       IF (56-LINES)40,40,41
   40  KEY=1
       GO TO 42
   41  CONTINUE
   22  KEY=0
       IF (N-NDIST)25,124,124
   25  IF (STP)10,10,12
  124  IF(NPROB-NP)125,125,9
  125  GO TO 23
       END


       SUBROUTINE RADOS1
       DIMENSION GMU(4),X(8)
       DIMENSION H(8),EN(4)
       DIMENSION SUMGP(4),THETA(4)
       DIMENSION GAM(8,8,4),DINT(8,8,4)
       DIMENSION BINT(8,8,4)
       COMMON STKHT,PI,UPLIM,ALPH,BETA,GMU,X,H,EN,SUMGP,NG,FLID
C
C
       DBLN2(B,BSQ,EN)=(2.*BEKI3(B)/(BSQ+1.)+(B-B/(BSQ+1.))*BKO(B)+((BSQ)
      1/(BSQ+1.)+BSQ/(7.*EN**2.4))*BK1(B))
C
       ALPH2=ALPH*ALPH
       BETA2=BETA*BETA
       CON=(2.*ALPH**2*BETA)/PI**2
       A=0.
       B=PI
       C=0.
       G=UPLIM
       V=0.5*(B-A)
       Q=0.5*(B+A)
       R=0.5*(G-C)
       W=0.5*(G+C)
       M=8
       DO 10 I=1,M
       DO 10 J=1,M
```

```
      THETA(1)=V*X(J)+Q
      THETA(2)=-V*X(J)+Q
      THETA(3)=THETA(1)
      THETA(4)=THETA(2)
      GAM(I,J,1)=R*X(I)+W
      GAM(I,J,2)=GAM(I,J,1)
      GAM(I,J,3)=-R*X(I)+W
      GAM(I,J,4)=GAM(I,J,3)
      DO 10 K=1,4
   10 DINT(I,J,K)=EXP(-0.5*ALPH2*(BETA2*GAM(I,J,K)**2*SIN(THETA(K))**2
     1            +(GAM(I,J,K)*COS(THETA(K))-1.)**2))
      DO 14 N=1,NG
      SUM=0.
      DO 11 I=1,M
      DO 11 J=1,M
      DO 11 K=1,4
      BARG=GMU(N)*GAM(I,J,K)*STKHT
      BASQ=BARG*BARG
      IF(BARG-174.)8,11,11
    8 BINT(I,J,K)=DBLN2(BARG,BASQ,EN(N))
      SUM=SUM+BINT(I,J,K)*DINT(I,J,K)*H(I)*H(J)
   11 CONTINUE
   14 SUMGP(N)=SUM*V*R*CON
      RETURN
      END


      SUBROUTINE RADOS2
      DIMENSION GMU(4),X(8)
      DIMENSION H(8),EN(4)
      DIMENSION SUMGP(4),SM(2)
      DIMENSION THETA(4),GMM(2)
      DIMENSION T(2),P(2)
      DIMENSION GAM(8,8,4),BINT(8,8,4)
      DIMENSION DINT(8,8,4)
      COMMON STKHT,PI,UPLIM,ALPH,BETA,GMU,X,H,EN,SUMGP,NG,FLID
      DBLN2(B,BSQ,EN)=(2.*BEKI3(B)/(BSQ+1.)+(B-B/(BSQ+1.))*BKO(B)+((BSQ)
     1/(BSQ+1.)+BSQ/(7.*EN**2.4))*BK1(B))
      DBLN1(BET,GAM,TH)=EXP(-(BET*(GAM*SIN(TH))**2.))
      BET=BETA*BETA/2.
      ALPH2=ALPH*ALPH
      BETA2=BETA*BETA
      G1=0.
      G2=1.0
      F1=0.
      F2=PI/2.
      CON=(2./PI)**1.5*BETA
      V=0.5*(F2-F1)
      Q=0.5*(F2+F1)
      R=0.5*(G2-G1)
      W=0.5*(G2+G1)
      M=8
      DO 10 I=1,M
      DO 10 J=1,M
      THETA(1)=V*X(J)+Q
      THETA(2)=-V*X(J)+Q
      THETA(3)=THETA(1)
      THETA(4)=THETA(2)
```

```
      GAM(I,J,1)=R*X(I)+W
      GAM(I,J,2)=GAM(I,J,1)
      GAM(I,J,3)=-R*X(I)+W
      GAM(I,J,4)=GAM(I,J,3)
      DO 10 K=1,4
   10 DINT(I,J,K)=DBLN1(BET,GAM(I,J,K),THETA(K))
      DO 14 N=1,NG
      SUM=0.
      DO 11 I=1,M
      DO 11 J=1,M
      DO 11 K=1,4
      BARG=GMU(N)*GAM(I,J,K)*FLID
      BASQ=BARG*BARG
      IF(BASQ-174.)8,11,11
    8 BINT(I,J,K)=DBLN2(BARG,BASQ,EN(N))
      SUM=SUM+BINT(I,J,K)*DINT(I,J,K)*H(I)*H(J)
   11 CONTINUE
   14 SUMGP(N)=SUM*V*R*CON
      G1=1.
      G2=UPLIM
      RR=0.5*(G2-G1)
      SS=0.5*(G2+G1)
      DO 60 N=1,NG
      SUM=0.
      DO 59 I=1,M
      GMM(1)=SS+RR*X(I)
      GMM(2)=SS-RR*X(I)
      FP=ATAN(SQRT(GMM(1)**2-1.))
      FM=ATAN(SQRT(GMM(2)**2-1.))
      P(1)=0.5*(F2-FP)
      T(1)=0.5*(F2+FP)
      P(2)=0.5*(F2-FM)
      T(2)=0.5*(F2+FM)
      DO 56 K=1,2
      SUM1=0.
      DO 58 J=1,M
      TH1=T(K)+P(K)*X(J)
      TH2=T(K)-P(K)*X(J)
   58 SUM1=SUM1+H(J)*(DBLN1(BET,GMM(K),TH1)+DBLN1(BET,GMM(K),TH2))
   56 SM(K)=SUM1*P(K)
      DO 59 K=1,2
      BARG=GMU(N)*GMM(K)*FLID
      BSQ=BARG*BARG
      IF(BSQ-174.)61,59,59
   61 SUM=SUM+H(I)*DBLN2(BARG,BSQ,EN(N))*SM(K)
   59 CONTINUE
   60 SUMGP(N)=SUMGP(N)+SUM*RR*CON
      RETURN
      END
```

```
      FUNCTION BI0(X)
      IF (X-3.75)2,2,3
    2 Z=(X/3.75)**2
      BI0=1.0+Z*(3.515623E0+Z*(3.089942E0+Z*(1.206749E0+Z*(.2659732E0
     1     +Z*(.360768E-1+Z*.45813E-2)))))
      GO TO 4
    3 Z=3.75/X
      BI0=(EXP(X)/SQRT(X))*(.3989423E0+Z*(.1328592E-1+Z*(.2253187E-2
     1     +Z*(-.1575649E-2+Z*(.9162808E-2+Z*(-.2057706E-1+Z*(.2635537E-1
     2     +Z*(-.1647633E-1+Z*.3923767E-2)))))))
    4 RETURN
      END


      FUNCTION BI1(X)
C
      IF (X-3.75)2,2,3
    2 Z=(X/3.75)**2
      BI1=(.5+Z*(.8789059E0+Z*(.5149887E0+Z*(.1508493E0+Z*(.2658733E-1
     1     +Z*(.301532E-2+Z*.32411E-3)))))*X
      GO TO 4
    3 Z=3.75/X
      BI1=(EXP(X)/SQRT(X))*(.3989423E0+Z*(-.3988024E-1+Z*(-.3620183E-2
     1     +Z*(.1638014E-2+Z*(-.1031555E-1+Z*(.2282967E-1+Z*(-.2895312E-1
     2     +Z*(.1787654E-1-Z*.4200587E-2)))))))
    4 RETURN
      END


      FUNCTION BK0(X)
C
      IF (X-2.0)2,2,3
    2 Z=(X/2.0)**2
      SERIES=-.5772157E0+Z*(.4227842+Z*(.2306976E0+Z*(.348859E-1
     1       +Z*(.262698E-2+Z*(.1075E-3+Z*.74E-5)))))
      BK0=SERIES-ALOG(X/2.0)*BI0(X)
      GO TO 4
    3 Z=2.0/X
      BK0=(1.253314E0+Z*(-.7832358E-1+Z*(.2189568E-1+Z*(-.1062446E-1
     1   +Z*(.587872E-2+Z*(-.25154E-2+Z*.53208E-3))))))/(SQRT(X)*EXP(X))
    4 RETURN
      END


      FUNCTION BK1(X)
C
      IF (X-2.0)2,2,3
    2 Z=(X/2.0)**2
      SERIES=1.+Z*(.1544314E0+Z*(-.6727858E0+Z*(-.181569E0+Z*(-.1919402E
     1      -1+Z*(-.110404E-2-Z*.4686E-4)))))
      BK1=SERIES/X+ALOG(X/2.0)*BI1(X)
      GO TO 4
    3 Z=2.0/X
      BK1=(1.253314+Z*(.2349862+Z*(-.365562E-1+Z*(.1504268E-1+Z*(-.
     1     780353E-2+Z*(.325614E-2-Z*.68245E-3))))))/(SQRT(X)*EXP(X))
    4 RETURN
      END
```

```
      FUNCTION BEKI3(X)
      IF (X-.1)1,2,2
    1 BEKI3=.7366554/(.9379389+X*(1.194192+X*(.5882452+X*(.5703372+X*
     1       (-1.579117+X*4.292469)))))
      GO TO 9
    2 IF (X-.4)3,4,4
    3 BEKI3=.5714978/(.7276787+X*(.9254691+X*(.4741521+X*(.2508204+X*
     1       (-.2593008E-1+X*.55708E-1)))))
      GO TO 9
    4 IF (X-1.)5,6,6
    5 BEKI3=.3272474/(.4166741+X*(.5295655+X*(.2754273+X*(.1283775+X*
     1       (.1191915E-1+X*.1392095E-1)))))
      GO TO 9
    6 IF (X-2.5)7,8,8
    7 BEKI3=(.221594+X*(-.9388379E-1+X*(.1473822E-1-X*.85765E-3)))/
     1       (.2826724+X*(.235632+X*(.6340205E-1+X*.1360032E-1)))
      GO TO 9
    8 Y=1./(X+3.25)
      BEKI3=1.268446*SQRT(Y)*EXP(-X)/(1.012074+Y*(-.325432E-3+Y*
     1       (-1.164632+Y*(1.387386-Y*4.465521))))
    9 RETURN
      END
```

RADOS SAMPLE PROBLEM, 2 GROUP, 1 ISOTOPE, 12 DISTANCES (INPUT IN CM)

RELEASE INVENTORY	PARENT,MEV/S	DECAY CONST,/S	ENERGY GROUP	DAUGHTER,MEV/S	DECAY CONST,/S	ENERGY GROUP
1	1.290000E 10	6.949999E-05	1.000000E 00	2.120000E 07	6.489998E-04	2.000000E 00

ENERGY GROUP CONSTANTS	DOSE CONV	ATTEN COEFF	ENERGY,MEV/S
1	4.499998E-10	6.149999E-05	2.000000E 00
2	5.719998E-10	1.340000E-04	5.000000E-01

	STACK HT	CY-SIGY	CZ-SIGZ	N	UBAR	GP-1 INT	GP-2 INT	GP-3 INT	GP-4 INT	DOSE	DISTANCE
1	2.150E 04	1.300E 04	1.400E 04	0.0	3.500E 02	6.074E-01	5.536E-01	0.0	0.0	1.195E-07	6.000E 04
2	2.150E 04	2.050E 04	2.150E 04	0.0	3.500E 02	4.764E-01	4.290E-01	0.0	0.0	9.462E-08	1.000E 05
3	2.150E 04	2.950E 04	3.150E 04	0.0	3.500E 02	3.164E-01	2.763E-01	0.0	0.0	6.337E-08	1.500E 05
4	2.150E 04	4.700E 04	4.950E 04	0.0	3.500E 02	1.594E-01	1.333E-01	0.0	0.0	3.224E-08	2.500E 05
5	2.150E 04	5.600E 04	5.850E 04	0.0	3.500E 02	1.194E-01	9.832E-02	0.0	0.0	2.421E-08	3.000E 05
6	2.150E 04	7.100E 04	7.500E 04	0.0	3.500E 02	7.766E-02	6.303E-02	0.0	0.0	1.578E-08	4.000E 05
7	2.150E 04	8.700E 04	9.100E 04	0.0	3.500E 02	5.385E-02	4.320E-02	0.0	0.0	1.092E-08	5.000E 05
8	2.150E 04	1.600E 05	1.200E 05	0.0	3.500E 02	2.299E-02	1.822E-02	0.0	0.0	4.466E-09	1.000E 06
9	2.150E 04	2.500E 05	1.200E 05	0.0	3.500E 02	1.477E-02	1.168E-02	0.0	0.0	2.616E-09	1.600E 06
10	2.150E 04	3.000E 05	1.200E 05	0.0	3.500E 02	1.232E-02	9.737E-03	0.0	0.0	2.030E-09	2.000E 06
11	2.150E 04	3.800E 05	1.200E 05	0.0	3.500E 02	9.735E-03	7.690E-03	0.0	0.0	1.459E-09	2.500E 06
12	2.150E 04	7.000E 05	1.200E 05	0.0	3.500E 02	5.289E-03	4.176E-03	0.0	0.0	4.841E-10	5.000E 06

IHC217I TRACEBACK FOLLOWS- ROUTINE ISN REGISTER 14

IBCOM 100 8E00D14C

MAIN 046 000049D8

LOAD POINT = 0000C820

664212

E. I. DU PONT DE NEMOURS & COMPANY
INCORPORATED
SAVANNAH RIVER LABORATORY
AIKEN, SOUTH CAROLINA 29801
(TWX: 803-824-0018, TEL: 803-824-6331, WU: AUGUSTA, GA.)

October 20, 1967

TO: Recipients of DP-1098, RADOS, A Code to Estimate Gamma Dose from a Cloud of Radioactive Gases, by R. E. Cooper

CORRECTION NOTICE

Please make the following corrections in your copy of DP-1098:

Page 9, Last paragraph relationships, Eq (11) denominators should be $\sqrt{2}$

Page 10, Eq (12) $S_{p,i}(t)$

Page 11, Eq (13) $S_{d,i}(t)$, $\lambda_{d,i}$ should be $\lambda_{p,i}$ in numerator

Page 16, Delete the following FORTRAN statements:

```
      SP(NI) = P(NI)*PLT
   14 SD(NI) = D(NI)*DLT + (P(NI)*(EN(NDD)*DL(NI)/(EN(NDP)*(DL(NI)-PL(NI))))
     1 *(PLT-DLT)
```

Replace with the following statements:

```
      IF(NDP.NE.0) GØ TØ 116
      SP(NI) = 0.
      SD(NI) = D(NI)*DLT + P(NI)*PL(NI)*EN(NDD)*(PLT-DLT)/(DL(NI)-PL(NI))
      GØ TØ 14
  116 SP(NI) = P(NI)*PLT
      IF(NDD.EQ.0) GØ TØ 33
      SD(NI) = D(NI)*DLT + P(NI)*PL(NI)*EN(NDD)*(PLT-DLT)/(EN(NDP)*(DL(NI)-
     1  PL(NI)))
      GØ TØ 14
   33 SD(NI) = 0.
   14 CONTINUE
```

These changes remove an error in statement 14 and in addition will allow the use of an isotope chain in which only the daughter is a gamma emitter. To utilize this feature, the following input requirements should be observed only for these particular chains:

(1) The parent source term should be entered as disintegrations per second rather than Mev/s.

(2) The parent decay energy group number should be entered as zero.

W. B. Scott
Technical Information Service

ERRATA

DP-1098, RADOS, A Code to Estimate Gamma Dose from a Cloud of Radioactive Gases, by R.E. Cooper

Page 9, Eg (11) Denominators should be $\sqrt{2}$ Last paragraph relationships

Page 10, Eg (12) $S_{p,i}(t)$

Page 11, Eg (13) $S_{d,i}(t)$, $\lambda_{d,i}$ should be $\lambda_{p,i}$ in numerator

Page 16, ~~Appendix A - RADOS main program~~

Delete ~~Remove~~ the following FORTRAN statements:

```
      SP(NI) = P(NI)*PLT
14    SD(NI) = D(NI)*DLT + (P(NI)*(EN(NDD)*DL(NI)/(EN(NDP)*(DL(NI)-PL(NI))))
     1 *(PLT-DLT)
```

Replace with the following statements:

```
      IF(NDP.NE.0) GØ TØ 116
      SP(NI) = 0.
      SD(NI) = D(NI)*DLT + P(NI)* PL(NI)*EN(NDD)*(PLT-DLT)/(DL(NI)-PL(NI))
      GØ TØ 14
116   SP(NI) = P(NI)*PLT
      IF(NDD.EQ.0) GØ TØ 33
      SD(NI) = D(NI)*DLT + P(NI)*PL(NI)*EN(NDD)*(PLT-DLT)/(EN(NDP)*(DL(NI)-
     1  PL(NI)))
      GØ TØ 14
33    SD(NI) = 0.
14    CONTINUE
```

These changes remove an error in statement 14 and in addition will allow the use of an isotope chain in which only the daughter is a gamma emitter. To utilize this feature, the following input requirements should be observed only for these particular chains:

(1) The parent source term should be entered as disintegrations per second rather than Mev/s.

(2) The parent decay energy group number should be entered as zero.

W. B. Scott